Math 3103 - Introduction to Differential Equations

Euler's Method for Systems

```
> restart;
  with(plots):
  with(DEtools):
```

Example system: x' = 2x(1-x/2) - xy   &   y' = 3y(1-y/3) - 2xy
(competing species)

First, we plot using Maple's tools.

Let's plot a direction field with some sample solutions satisfying initial conditions:
x(0)=y(0)=0.5, x(0)=y(0)=1.5, x(0)=1 & y(0)=3, x(0)=3 & y(0)=1.

```
> DEsys := [diff(x(t),t)=2*x(t)*(1-x(t)/2)-x(t)*y(t), diff(y(t),t)=3*y
  (t)*(1-y(t)/3)-2*x(t)*y(t)];
```

$$DEsys := \left[\frac{d}{dt} x(t) = 2\, x(t) \left(1 - \frac{1}{2}\, x(t)\right) - x(t)\, y(t),\ \frac{d}{dt} y(t) = 3\, y(t) \left(1 - \frac{1}{3}\, y(t)\right) - 2\, x(t)\, y(t) \right] \quad (1)$$

```
> DEplot(DEsys,[x(t),y(t)],t=0..5,{[0,0.5,0.5],[0,1.5,1.5],[0,1,3],[0,3,
  1]},x=0..3,y=0..3);
```



Next, let's define the right hand side of the equations as functions f and g and then solve f = 0 & g = 0 to find equilibria solutions.

```
> f := (t,x,y) -> 2*x*(1-x/2)-x*y:
  'f(t,x,y)' = f(t,x,y);

  g := (t,x,y) -> 3*y*(1-y/3)-2*x*y:
  'g(t,x,y)' = g(t,x,y);
```

$$f(t, x, y) = 2\,x\left(1 - \frac{1}{2}\,x\right) - x\,y$$

$$g(t, x, y) = 3\,y\left(1 - \frac{1}{3}\,y\right) - 2\,x\,y \tag{2}$$

```
> solve({f(t,x,y)=0,g(t,x,y)=0});
```

$$\{x=0, y=0\}, \{x=0, y=3\}, \{x=2, y=0\}, \{x=1, y=1\} \tag{3}$$

We have 4 equilibria: (0,0), (0,3), (2,0), and (1,1).

Finally, let's implement Euler's Method for systems (assuming a system with 2 equations / 2 dependent variables).

We'll assume our initial condition is x(5) = 1 and y(5) = 3.

Let's do 10 steps with step size 0.2 to approximate x(7) and y(7).

```
> # initial conditions
  t[0] := 5;
  x[0] := 1;
  y[0] := 3;

  # Number of interations and step size.
  N := 10;
  h := 0.2;

  # Euler's for 2-var systems
  for n from 0 to N-1 do
     t[n+1] := t[n] + h;

     x[n+1] := x[n] + f(t[n],x[n],y[n])*h;
     y[n+1] := y[n] + g(t[n],x[n],y[n])*h;
  end do;
```

$$t_0 := 5$$
$$x_0 := 1$$
$$y_0 := 3$$
$$N := 10$$
$$h := 0.2$$
$$t_1 := 5.2$$
$$x_1 := 0.6$$
$$y_1 := 1.8$$
$$t_2 := 5.4$$
$$x_2 := 0.5520000000$$
$$y_2 := 1.8$$
$$t_3 := 5.6$$
$$x_3 := 0.5131392000$$
$$y_3 := 1.834560000$$
$$t_4 := 5.8$$
$$x_4 := 0.4774555821$$
$$y_4 := 1.885620061$$
$$t_5 := 6.0$$
$$x_5 := 0.4427850836$$
$$y_5 := 1.945759565$$
$$t_6 := 6.2$$
$$x_6 := 0.4083767287$$
$$y_6 := 2.011397922$$
$$t_7 := 6.4$$

$$x_7 := 0.3740914890$$
$$y_7 := 2.080529114$$
$$t_8 := 6.6$$
$$x_8 := 0.3400775493$$
$$y_8 := 2.151803010$$
$$t_9 := 6.8$$
$$x_9 := 0.3066220423$$
$$y_9 := 2.224121620$$
$$t_{10} := 7.0$$
$$x_{10} := 0.2740745012$$
$$y_{10} := 2.296465311 \quad \textbf{(4)}$$